

Fixed Point Iteration: Logistic Function

Mth 351 June 30 2002 Maple 6

Bent E. Petersen

Filename: 351u2002_logistic.mws

Assignment 1. Due July 8, 2002. See the two problems at the end of this worksheet.

```
> restart;
```

We will investigate the fixed point iteration for the logistic function

```
> f := (x, c) -> c*x*(1-x);
```

$$f := (x, c) \rightarrow cx(1-x)$$

Here c is a parameter which satisfies $0 \leq c \leq 4$. For these values of c we have $0 \leq f(x,c) \leq 1$ for each x satisfying $0 \leq x \leq 1$, that is, as a function of x , $f(x,c)$ maps the interval $[0,1]$ into itself. Thus by the Intermediate Value Theorem (IVT) $f(x,c)$ is guaranteed to have a fixed point in $[0,1]$. This fact is not particularly useful here though since the origin is an obvious fixed point, and the IVT doesn't give us any additional fixed points.

It is not difficult to see that $f(x,c)$ is a contraction map for $0 \leq c < 1$. In this case we have a unique fixed point and the fixed point iterations will converge very rapidly to the unique fixed point, no matter what initial "guess" we start with. For other values of c it is not clear how the fixed point iterates will behave.

Our goal here is to plot some iterates for various values of c and attempt to describe what we observe. Even in the case of convergence to a fixed point, the first few iterates may depend heavily on the initial point and so may be all over the place. Therefore we discard the first M_0 iterates and plot only subsequent iterates.

Here are the values of the parameter c that we will work with. We omit 0 because it is not interesting.:

```
> N0 := 30; # N0 values of c
```

```
N0 := 30
```

```
> Cvals := [seq(4*k/N0, k=1..N0)];
```

$$Cvals := \left[\frac{2}{15}, \frac{4}{15}, \frac{2}{5}, \frac{8}{15}, \frac{2}{3}, \frac{4}{5}, \frac{14}{15}, \frac{16}{15}, \frac{6}{5}, \frac{4}{3}, \frac{22}{15}, \frac{8}{5}, \frac{26}{15}, \frac{28}{15}, 2, \frac{32}{15}, \frac{34}{15}, \frac{12}{5}, \frac{38}{15}, \frac{8}{3}, \frac{14}{5}, \frac{44}{15}, \frac{46}{15}, \frac{16}{5}, \frac{10}{3}, \frac{52}{15}, \frac{18}{5}, \frac{56}{15}, \frac{58}{15}, 4 \right]$$

Here are the initial points. We omit 0 because it is a fixed point and so it's not very exciting.


```
> M0:=16; # M0 values of x0
                                M0 := 16
> Inits:=[seq(k/M0, k=1..M0)];
                                Inits :=  $\left[ \frac{1}{16}, \frac{1}{8}, \frac{3}{16}, \frac{1}{4}, \frac{5}{16}, \frac{3}{8}, \frac{7}{16}, \frac{1}{2}, \frac{9}{16}, \frac{5}{8}, \frac{11}{16}, \frac{3}{4}, \frac{13}{16}, \frac{7}{8}, \frac{15}{16}, 1 \right]$ 
> Q0:=3; # Number of iterates to discard
                                Q0 := 3
> R0:=25; # R0 is number of iterates to plot
                                R0 := 25
```

We loop over the parameter values, c, and the initial points, x0, and for each initial point x0 we do Q0+R0 iterates and discard the first Q0+1 iterates (including x0). The remaining iterates are saved in a list of M0*R0*N0 points.

```
> L:=[]: # list of all points
> prec:=6: # precision
> for k from 1 to nops(Cvals) do
>   c:=evalf(Cvals[k],prec):
>   for j from 1 to nops(Inits) do
>     x0:=evalf(Inits[j],prec):
>     for h from 1 to Q0 do
>       x0:=evalf(f(x0,c),prec):
>     od:
>     for h from 1 to R0 do
>       x0:=evalf(f(x0,c),prec): # compute new point
>       L:=[op(L), [c,x0]]: # push on L
>     od:
> od: od:
```

Let's plot the points in the list L. Here's the number of points we will be plotting:

```
> nops(L); M0*R0*N0;
                                12000
                                12000
> PLOT(POINTS(op(L)), SYMBOL(DIAMOND, 12));
```



We can see another fixed point (other than 0) emerging when $c > 1$. When $c > 3$ (or so) the behavior of the iterates changes dramatically.

Here are some problems for you. You can do them "by hand."

Problem 1. When $1 < c \leq 4$ there are two fixed points. What are they (in terms of c).

Problem 2. Use initial values $1/4, 1/2, 3/4$, compute a few iterates and comment on their apparent behavior. Do this for $c = 2, 3, 3.9$.

[>